

Table of Contents

Workspace Recipes	1
Creating Workspaces	1
Creating a Directory Workspace	2

Workspace Recipes

The Workspace classes are in the [geoscript.workspace](#) package.

A Workspace is a collection of Layer. You can create, add, remove, and get Layers. There are many different kinds of Workspace in GeoScript including Memory, PostGIS, Directory (for Shapefiles), GeoPackage, and many more.

Creating Workspaces

Create a Workspace

```
Workspace workspace = new Workspace()
```

Create a Layer

```
Schema schema = new Schema("cities", [  
    new Field("geom", "Point", "EPSG:4326"),  
    new Field("id", "Integer"),  
    new Field("name", "String")  
)  
Layer layer = workspace.create(schema)  
println layer
```

```
cities
```

Check whether a Workspace has a Layer by name

```
boolean hasCities = workspace.has("cities")  
println hasCities
```

```
true
```

Get a Layer from a Workspace

```
Layer citiesLayer = workspace.get('cities')  
println citiesLayer
```

```
cities
```

Add a Layer to a Workspace

```
Schema statesSchema = new Schema("states", [  
    new Field("geom", "Polygon", "EPSG:4326"),  
    new Field("id", "Integer"),  
    new Field("name", "String")  
])  
Layer statesLayer = new Layer("states", statesSchema)  
workspace.add(statesLayer)  
println workspace.has("states")
```

true

Get the names of all Layers in a Workspace

```
List<String> names = workspace.names  
names.each { String name ->  
    println name  
}
```

cities
states

Remove a Layer from a Workspace

```
workspace.remove("cities")  
println workspace.has('cities')
```

false

Close the Workspace when you are done

```
workspace.close()
```

Creating a Directory Workspace

A Directory Workspace is a directory of Shapefiles.

Create a Directory Workspace

```
Directory directory = new Directory("src/main/resources/data")  
println directory.toString()
```

```
Directory[/Users/jericks/Projects/geoscript-cookbook/src/main/resources/data]
```

View the Workspace's format

```
String format = directory.format  
println format
```

```
Directory
```

View the Workspace's File

```
File file = directory.file  
println file
```

```
/Users/jericks/Projects/geoscript-cookbook/src/main/resources/data
```

View the Workspace's list of Layer names

```
List names = directory.names  
names.each { String name ->  
    println name  
}
```

```
states
```

Get a Layer by name

```
Layer layer = directory.get("states")  
int count = layer.count  
println "Layer ${layer.name} has ${count} Features."
```

```
Layer states has 49 Features.
```

Close the Directory when done.

```
directory.close()
```